query: SELECT ?item ?itemLabel ?image ?article ?image_title
  WITH
  {
    SELECT ?item ?image ?article ?image_title
    WHERE 
    {
      ?article schema:about ?item ;
               schema:isPartOf <https://vi.wikipedia.org/> .
      ?item wdt:P31 wd:Q16521 .
      ?item wdt:P18 ?image .
      ?item wdt:P5984 ?apni .
      FILTER (CONTAINS(STR(?image),"FilePath/P"))
      FILTER (CONTAINS(STR(?article),"wiki/P"))  
      BIND (SUBSTR(wikibase:decodeUri(STR(?image)), 52) AS ?image_title)
    }
  } AS %i
  WITH
  {
    SELECT ?image_title
    WHERE
    {
      SERVICE wikibase:mwapi
      {
        bd:serviceParam wikibase:endpoint "commons.wikimedia.org" .
        bd:serviceParam wikibase:api "Generator" .
        bd:serviceParam mwapi:generator "categorymembers" .
        bd:serviceParam mwapi:gcmtitle "Category:Photographs by Kevin Thiele" .
        bd:serviceParam mwapi:gcmtype "file" .
        bd:serviceParam mwapi:gcmprop "title" .
        bd:serviceParam mwapi:gcmlimit "max" .
        ?file wikibase:apiOutput mwapi:title .
      }
      BIND (SUBSTR(?file, 6) AS ?image_title)
    }
  } AS %wanted_image_titles
  WHERE
  {
    INCLUDE %i
    INCLUDE %wanted_image_titles
    FILTER NOT EXISTS
    { 
      ?article2 schema:about ?item ;
                schema:isPartOf <https://en.wikipedia.org/> .
    }
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
  }